

## TI83 Quadratic Formula Program QUADRATC

```
:ClrHome
:Full
:Func
:Float
:1→Xscl
:1→Yscl
:Output(1,1,"GRAPHICAL"
:Output(2,4,"QUADRATIC"
:Output(3,8,"EQUATIONS"
:Output(5,1,"AX2 + BX + C = 0"
:Output(7,3,"TO CONTINUE,"
:Output(8,3,"PRESS ENTER."
:Pause
:ExprOff
:FnOff
: "AX2+BX+C"→Y1
:Lbl 1
:ClrHome
:Output(7,4,"ENTER THE"
:Output(8,3,"COEFFICIENTS"
:Disp "AX2 + BX + C = 0"
:Input "A=",A
:If A= 0:Then
:Output(5,1,"A CANNOT BE ZERO."
:Output(7,3,"PRESS ENTER"
:Output(8,2,"AND TRY AGAIN."
:Pause
:Goto 1
:End
:Input "B=",B
:Input "C=",C
:B2-4A*C→D
:Output(6,1,"DISCRIM = B2-4AC"
:Output(7,1,"="
:Output(7,3,D
:Output(8,1,"* PRESS ENTER *"
:Pause
:ClrHome
:-B/(2A)→X
:If A>0:Then
:Y1→Ymin
:Y1+10→Ymax
:End
:If A<0:Then
```

Given the coefficients of a quadratic equation, it gives the solutions, real or complex, graphs the corresponding function, gives the vertex, and allows you to trace along the graph.

```

:Y1→Ymax
:Y1-10→Ymin
:End
:If Ymin>-1:-5→Ymin
:If Ymax<1:5→Ymax
:X-7→Xmin
:X+7→Xmax
:If D>0:Then
:(-B+√(D))/(2A)→U
:(-B-√(D))/(2A)→L
:If A<0:Then
:U→W
:L→U
:W→L
:End
:U+4→Xmax
:L-4→Xmin
:End
:If Xmin≥0:-2→Xmin
:If Xmax≤0:2→Xmax
:Horiz
:Fix 5
:If D>0:Then
:Outpt(1,1,"2 REAL SOLNS:")
:Outpt(2,1,"X = "
:Outpt(2,15, "OR"
:Outpt(2,5,L
:Outpt(3,1,"X = "
:Outpt(3,5,U
:End
:If D=0:Then
:-B/(2A)→U
:Outpt(1,1,"1 REAL SOLUTION:")
:Outpt(2,1,"X = "
:Outpt(2,5,U
:End
:If D<0:Then
:-B/(2A)→R
:abs(√(-D)/(2A))→I
:Outpt(1,1,"2 COMPLEX SOLNS:")
:Outpt(2,1,"X = "
:Outpt(2,5,R
:Outpt(3,3,"+/- "
:Outpt(3,8,I
:Outpt(3,16,"i")

```

```

:End
:Output(4,1,"* PRESS ENTER *")
:Pause
:ClrHome
:-B/(2A)→X
:Output(1,1,"PARABOLA VERTEX:")
:Output(2,1,"X = "
:Output(2,5,X
:Output(3,1,"Y = "
:Output(3,5,Y1
:Output(4,1,"* PRESS ENTER *")
:Pt-On(X,Y1,2
:Pause
:Full
:Float
:ClrHome
:Lbl 2
:Menu("CHOOSE AN OPTION", "NEW PROBLEM",1,"TRACE GRAPH",3,"QUIT",4)
:Lbl 3
:(Xmin+Xmax)/2→M
:2^int(ln((Xmax-Xmin)/94)/ln(2)+.5)→K
:int(M-47K+.5)→Xmin
:K→Δx
:Ymin-.12(Ymax-Ymin)→Ymin
:Trace
:Goto 2
:Lbl 4
:FnOff
:ExprOn
:Stop

```

**ClrHome** clears the home screen. It is found under the **I/O** menu within the **Program** menu (obtained through pressing the **PRGM** button while editing a program). Also under the **I/O** menu, you will find **Input**, **Output**, and **Disp**.

The **quote marks** are the **ALPHA** function of the **PLUS** sign.

The **CTL** (control) menu within the **Program** menu contains **If**, **Then**, **Else**, **End**, **Menu**, **Lbl**, **Goto**, **Pause**, and **Stop**.

**Func** sets the calculator mode to graph functions in  $x$  and  $y$ . **Func** is gotten by selecting it in the **MODE** menu. Also, in the **MODE** menu, you will find **Float**. This sets the data output to use as many decimals as it needs, as opposed to rounding all answers to a specific number of decimal places.

**Fix** sets the number of decimal places in outputs to be a specific number. To get **Fix 5**, while editing the program, select **MODE** and select **5** in the line “**Float 012345678901**”.

Most objects are found in the **Catalog**. The catalog menu (**CATALOG**) is the second function of the **0** button. Once in the **CATALOG** menu, practically everything can be gotten. However, it may be easier to search the specific menus.

To access **Xmin**, **Xmax**, **Xscl**, **Ymin**, **Ymax**, **Yscl**, and  $\Delta x$ , press the **VAR**S button and select **Window**.

The **equal sign** is under the **TEST** menu, which is the second function of the **MATH** button. The **colon :** is the **ALPHA** function of the **decimal point**. The **comma ,** has its own button located above the 7 key.

The **inequality signs** are found under the **TEST** menu, the second function of the **MATH** button.

**Y<sub>1</sub>** is obtained by pressing the **VAR**S button, then selecting **Y-VARS** at the top. Then select **Function**. **Y1** will be an option under **Function**.

You can also find **FnOff** here under **VAR**S. Once you press **VAR**S, select **Y-VARS**, and then select **On/Off**. From this menu, select **FnOff**.

**ExprOff** is found under the **FORMAT** menu, the second function of the **ZOOM** button. Simply enter this menu when you are in the appropriate place in the program. Then arrow down and over to **ExprOff**. Press **ENTER** once you are over the appropriate expression. This should put the expression into your program.

**Full** and **Horiz** are found in the **MODE** menu. Again, start from the appropriate place in the program. Then press the **MODE** button and arrow over and down to **Horiz** and **Full**.

**Trace** is gotten by simply pressing the **TRACE** button in the appropriate place in the program.

**Pt-On** can be found under the **DRAW** menu, which is the second function of the **PRGM** button. Select **POINTS** along the top, then **Pt-On**( from the list.

The  $\rightarrow$  arrow is gotten pressing the **STO $\rightarrow$**  button to the left of the number pad.

The **X** is gotten by the regular **X,T,q,n** button you use when graphing.

**int** and **abs** are found under the **NUM** menu within the **MATH** menu. The **MATH** menu is accessed by pressing the **MATH** button.

The **ALPHA** function of the **0** button in the number pad is the **space** key. It looks like a little half rectangle on its side.

**ln** has its own button. It looks like **LN** and it's on the left of the calculator.

The **asterisk** **\*** is the multiplication button.